;Save PDF to dwg
;Start in a new, empty drawing
;Select PDF file
;The program imports all pages and save each page in a separate dwg file
;DWG names are "<PDF file> Page#.dwg"
;Stefan M. - for Joseph Meyer, via Upwork
;  v1.01 - 27.04.2020
(defun c:allpdf ( / *error* acdoc expert
                     file i n dir ss
                     en el o d ob
                     dwg overwrite k wb
                )
  (vl-load-com)
  (setq acdoc (vla-get-activedocument (vlax-get-acad-object)))

  (if
    (= 8 (logand (getvar 'undoctl) 8))
    (vla-endundomark acdoc)
  )
  (vla-startundomark acdoc)
  (setq expert (getvar 'expert))
  (setvar 'expert 2)
  
  (defun *error* (msg)
    (and msg
      (not (wcmatch (strcase msg) "*CANCEL*,*QUIT*,*BREAK*,*EXIT*"))
      (princ (strcat "\nError: " msg))
    )
    (setvar 'expert expert)
    (if
      (= 8 (logand (getvar 'undoctl) 8))
      (vla-endundomark acdoc)
    )
    (princ)
  )

  (setvar 'pdfimportfilter 8)
  (setvar 'pdfimportmode  22)
  (setvar 'pdfimportlayers 0)
  

  (foreach la '(("TEXT" . 1) ("LINESSHORTS" . 5) ("LINESMEDIUM" . 3) ("LINESGRID" . 2))
    (or
      (tblsearch "layer" (car la))
      (entmakex
        (list
          '(0 . "LAYER")
          '(100 . "AcDbSymbolTableRecord")
          '(100 . "AcDbLayerTableRecord")
          (cons 2 (car la))
          '(70 . 0)
          (cons 62 (cdr la))
          '(6 . "Continuous")
          '(290 . 1)
          '(370 . -3)
        )
      )
    )
  )

  (setq i 0 n 0)

  (if
    (setq ss (ssget "_X"))
    (alert "\nDrawing is not empty.\nOpen a new, empty drawing and try again.")
    (progn
      (if
        (setq file (getfiled "Select PDF file to import" (getvar 'dwgprefix) "pdf" 2))
        (progn
          (setq dir (strcat (vl-filename-directory file) "\\")
                name (vl-filename-base file)
          )
          (while
            (progn
              (vl-cmdf "-pdfimport" "_file" file (setq i (1+ i)))
              (if (wcmatch (getvar 'lastprompt) (strcat "*" file "*"))
                (progn (command) nil)
                (progn (vl-cmdf '(0 0) 1.0 0.0) T)
              )
            )
            (if
              (setq ss (ssget "_X"))
              (progn
                (repeat (setq j (sslength ss))
                  (setq en (ssname ss (setq j (1- j)))
                        el (entget en)
                        ob (vlax-ename->vla-object en)
                        o (cdr (assoc 0 el))
                  )
                  (cond
                    ((wcmatch o "*TEXT")
                     (vla-put-layer ob "TEXT")
                     (vla-put-color ob 256)
                    )
                    ((wcmatch o "LINE,*POLYLINE")
                     (setq d (vlax-curve-getdistatparam en (vlax-curve-getendparam en)))
                     (cond
                       ((<= d (/ 1.0 64.0))
                        (vla-put-layer ob "LINESSHORTS")
                        (vla-put-color ob 256)
                       )
                       ((<= d (/ 3.0 64.0))
                        (vla-put-layer ob "LINESMEDIUM")
                        (vla-put-color ob 256)
                       )
                       ((not (wcmatch (strcase (vla-get-linetype ob)) "BYLAYER,CONTINUOUS"))
                        (vla-put-layer ob "LINESGRID")
                        (vla-put-color ob 256)
                       )
                     )
                    )
                    ((or
                       (and (eq o "HATCH") (eq (strcase (cdr (assoc 2 el))) "SOLID"))
                       (eq o "SOLID")
                     )
                     (ssdel en ss)
                     (entdel en)
                    )
                  )
                )
                (if
                  (> (sslength ss) 0)
                  (progn
                    (setq dwg (strcat dir name "_Page " (itoa i) ".dwg"))
                    (if
                      (and
                        (findfile dwg)
                        (not overwrite)
                      )
                      (progn
                        (initget "Yes No yesAll" 1)
                        (setq k (getkword (strcat "\n" dwg " exist. Overwrite? [Yes/No/yesAll]: ")))
                        (cond
                          ((eq k "Yes") (setq wb T))
                          ((eq k "No") (setq wb nil))
                          ((eq k "yesAll") (setq overwrite T))
                        )
                      )
                      (setq wb T)
                    )
                    (if
                      (or overwrite wb)
                      (progn
                        (vl-cmdf "_.-wblock"  dwg "" "_non" '(0.0 0.0 0.0) ss "")
                        (setq n (1+ n))
                      )
                      (repeat (setq j (sslength ss)) (entdel (ssname ss (setq j (1- j)))))
                    )
                  )
                )
              )
            )
          )
        )
      )
    )
  )
  (princ (strcat "\n" (itoa n) " Pages imported."))
  (*error* nil)
  (princ)
)
